/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */

plugins {
    id 'groovy'
    id 'java-library'
    id 'org.apache.grails.buildsrc.publish'
    id 'org.apache.grails.buildsrc.sbom'
}

version = projectVersion
group = 'org.apache.grails.data'

ext {
    pomTitle = 'Grails GORM'
    pomDescription = 'GORM - Grails Data Access Framework'
    pomDevelopers = [
            'graemerocher': 'Graeme Rocher',
            'jeffscottbrown': 'Jeff Brown',
            'burtbeckwith': 'Burt Beckwith',
            'jameskleeh': 'James Kleeh',
            'puneetbehl': 'Puneet Behl',
            'jamesfredley': 'James Fredley'
    ]
}

dependencies {
    implementation platform(project(':grails-bom'))
    compileOnly 'org.apache.groovy:groovy'
    compileOnly 'org.spockframework:spock-core'
    api project(':grails-datastore-core'), {
        // api: AbstractPersistenceEvent, AbstractPersistenceEventListener, AbstractQueryEvent, Datastore,
        //      DirtyCheckable, ProxyHandler, Session
        // impl: DatastoreUtils, EventType, OneToOne, OptimisticLockingException, PersistentEntity, PostDeleteEvent,
        //       PostQueryEvent, PreDeleteEvent, PreInsertEvent, PreQueryEvent, PreUpdateEvent, SessionCreationEvent,
        //       ValidationEvent
        if (excludeUnusedTransDeps)  {
            // API dependencies in grails-datastore-core
            //exclude group: 'jakarta.persistence', module: 'jakarta.persistence-api' // Entity
            exclude group: 'org.javassist', module: 'javassist'
            exclude group: 'org.springframework', module: 'spring-context'
            exclude group: 'org.springframework', module: 'spring-core'
            exclude group: 'org.springframework', module: 'spring-tx'
        }
    }

    api project(':grails-datamapping-core'), {
        // api: ApplyDetachedCriteriaTransform, DetachedCriteria, DirtyCheck, Entity
        // impl: CascadingValidator, ConfigurableApplicationEventPublisher, GroovyProxyFactory, PagedResultList
        if (excludeUnusedTransDeps) {
            // API dependencies in grails-datamapping-core
            exclude group: 'jakarta.annotation', module: 'jakarta.annotation-api'
            exclude group: 'jakarta.persistence', module: 'jakarta.persistence-api'
            exclude group: 'org.apache.grails.data', module: 'grails-datastore-core'
            //exclude group: 'org.apache.grails.data', module: 'grails-datamapping-validation' // CascadingValidator
            exclude group: 'org.springframework', module: 'spring-context'
            exclude group: 'org.springframework', module: 'spring-jdbc'
            exclude group: 'org.springframework', module: 'spring-tx'
        }
    }
    api project(':grails-datamapping-async'), {
        // api: AsyncEntity
        if (excludeUnusedTransDeps) {
            // API dependencies in grails-datamapping-async
            exclude group: 'org.apache.grails.data', module: 'grails-datamapping-core'
            //exclude group: 'org.apache.grails.async', module: 'grails-async' // PromiseDecoratorProvider
        }
    }
    api project(':grails-core'), {
        // impl: Entity, ValidationException
        if (excludeUnusedTransDeps) {
            // API dependencies in grails-core
            exclude group: 'jakarta.annotation', module: 'jakarta.annotation-api'
            exclude group: 'jakarta.inject', module: 'jakarta.inject-api'
            exclude group: 'jakarta.persistence', module: 'jakarta.persistence-api'
            //exclude group: 'org.apache.grails.bootstrap', module: 'grails-bootstrap' // Resource
            exclude group: 'org.apache.grails.data', module: 'grails-datastore-core'
            exclude group: 'org.apache.grails', module: 'grails-spring'
            exclude group: 'org.springframework', module: 'spring-beans'
            exclude group: 'org.springframework', module: 'spring-context'
            exclude group: 'org.springframework', module: 'spring-core'
            exclude group: 'org.springframework', module: 'spring-tx'
            exclude group: 'org.springframework.boot', module: 'spring-boot'
            exclude group: 'org.springframework.boot', module: 'spring-boot-autoconfigure'
        }
    }

    api 'org.springframework:spring-tx', {
        // impl: DataIntegrityViolationException
    }

    compileOnlyApi project(':grails-datamapping-support'), {
        // ast transforms
    }

    api 'org.apache.groovy:groovy-dateutil', {
        // Groovy Date Utils Extensions are used in the tests
    }
}

apply {
    from rootProject.layout.projectDirectory.file('gradle/java-config.gradle')
    from rootProject.layout.projectDirectory.file('gradle/code-style-config.gradle')
    from rootProject.layout.projectDirectory.file('gradle/docs-config.gradle')
    // do NOT do test configuration here, this is a TCK module
}
